// File: src/LPLocker.sol



/**                                
Nchart Token LP locker

Website: nchart.io
Docs: docs.nchart.io
twitter.com/Nchart_
twitter.com/Kekotron_
*/

// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "@openzeppelin/token/ERC20/IERC20.sol";

/**
            ........            
       ..::::::::::::.  .       
     .:::::::::::::::.  =+-.    
   --::::::::::::::::.  =+++-   
  *##*+::::::::::::::.  =+++++  
 *#####:  .::::::::::.  =++++++ 
-######:     .:::::::.  =++++++-
*######:  :.    .::::.  =+++++++
#######:  -=-:.    .:.  =+++++++
+######:  -=====:.      =++++++=
:######:  -========-.   =++++++:
 +#####:  -===========-.-+++++= 
  =####:  -==============-==+-  
   :*##:  -================-.   
     :+:  -==============-.     
          :==========-:.        
             ......                                       
*/

/**
 * @dev This contract allows locking of an ERC20 LP Token for a period 
 * of 2 years, and at some point after that, trigger a 90-day timer.
 * After the timer expires, the locked LP can be withdrawn.
 * No other action may be taken in the interim.
 */

contract LPLocker {
    address public owner;
    address public immutable tokenContract;
    uint256 public immutable earliestUnlockTime;
    uint256 public lockUpEndTime;
    bool public isLiquidityLocked;
    bool public isWithdrawalTriggered;

    error CannotAssignOwnerToAddressZero();
    error LockupNotEnded();
    error LPAlreadyLocked();
    error LPNotLocked();
    error OnlyOwnerCanCall();
    error TwoYearMinimum();
    error WithdrawalAlreadyTriggered();
    error WithdrawalNotTriggered();

    constructor(address tokenContract_) {
        earliestUnlockTime = block.timestamp + (365 days * 2);
        tokenContract = tokenContract_;
        owner = msg.sender;
    }

    function lockLiquidity(uint256 amount) external {
        _requireIsOwner();

        if (isLiquidityLocked) {
            revert LPAlreadyLocked();
        }

        IERC20(tokenContract).transferFrom(msg.sender, address(this), amount);

        isLiquidityLocked = true;
    }

    function triggerWithdrawal() external {
        _requireIsOwner();

        if(block.timestamp < earliestUnlockTime) {
            revert TwoYearMinimum();
        }

        if (!isLiquidityLocked) {
            revert LPNotLocked();
        }

        if (lockUpEndTime != 0) {
            revert WithdrawalAlreadyTriggered();
        }

        lockUpEndTime = block.timestamp + 90 days;
        isWithdrawalTriggered = true;
    }

    function cancelWithdrawalTrigger() external {
        _requireIsOwner();

        if (!isLiquidityLocked) {
            revert LPNotLocked();
        }

        if (lockUpEndTime == 0) {
            revert WithdrawalNotTriggered();
        }

        lockUpEndTime = 0;
        isWithdrawalTriggered = false;
    }

    function withdrawLP(uint256 amount) external {
        _requireIsOwner();

        if (!isLiquidityLocked) {
            revert LPNotLocked();
        }
        if (lockUpEndTime == 0) {
            revert WithdrawalNotTriggered();
        }
        if (block.timestamp < lockUpEndTime) {
            revert LockupNotEnded();
        }

        IERC20(tokenContract).transfer(owner, amount);

        isLiquidityLocked = false;
        lockUpEndTime = 0;
        isWithdrawalTriggered = false;
    }

    function changeOwner(address newOwner) external {
        _requireIsOwner();
        if (newOwner == address(0)) {
            revert CannotAssignOwnerToAddressZero();
        }
        owner = newOwner;
    }

    function _requireIsOwner() internal view {
        if (msg.sender != owner) {
            revert OnlyOwnerCanCall();
        }
    }
}

// File: lib/openzeppelin-contracts/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}